
hose of us who have been in the SAS world for a few 
years generally don't appreciate that we might be getting 
older.   It  was,  therefore,  difficult  for  me to  write  an 

obituary  for  well-known  and  well  respected  SAS  consultant, 
David Johnson, who died at an age close to mine.

T
The autumn SAS conference season has now ended, so those 

of us wanting to present  at  events in 2009 can concentrate  on 
actually writing their papers.  I've included reports from the 2008 
PhUSE  and  SAS  UK  events  in  this  issue.   As  well  as  the 
conference  reports  I  have  more  articles  from familiar  sources 
again in this issue of VIEWS News: including some advice on the 
use  of   horizontal  bar  charts  from  LeRoy  Bessler,  creating 
drill-down bubble plots from Phil Mason, and yet more tips on 
using  regular  expressions  from  the  Amadeus  Software 
Consultancy Team.  Along with our regular sections, including 
some new SAS Formats,  Options, and Functions in SAS 9.2,  I 
hope you will find at least some of them relevant to your use of 
SAS software.

If  you  would  like  to  contribute  an  article,  or  discuss  the 
possibility  of  doing  so,  please  send  an  email  to  me  at 
newsletter@views-uk.org.  A list of possible subjects for your 
articles can be found on the VIEWS web site, and anyone is very 
welcome to add  to  that  list  by sending emails  with your  own 
questions to newsletter@views-uk.org.

Philip R Holland (Newsletter Editor)

Membership of VIEWS is free and you can register for 
notification of all future VIEWS events by emailing us at  

membership@views-uk.org, making certain you include  
your name and postal address.  If you would like to receive 

email notification whenever a new issue of the free quarterly 
VIEWS News is published on the web site, please remember 

to also include your email address.

In Memoriam
David Johnson 1956-2008
David Johnson was an outstanding volunteer  and leader  in the 
SAS  user  conference  community.   Always  gracious  and 
conscientious,  David  was also  innovative  as  SUGI  conference 
section  chair.   He  was  a  pleasure  and  honour  to  serve  as  a 
presenter.  David will be missed.

LeRoy Bessler, Assurant Health

David Johnson was an excellent speaker (on almost any topic!), a 
well-respected SAS author, and a regular contributor to SAS-L. 
His advice and comments on SAS will be greatly missed.

Philip R Holland (Newsletter Editor) 
More memories of David Johnson can be found at

tinyurl.com/62ns9n

Did You Know?
If you have a useful hint or tip, please send it to the Editor,  
and share it with the VIEWS membership.

Show Them What Is Important:
Using a Horizontal Bar Chart Also As a Table
[SAS 8] Simple techniques to show readers what is important are 
ranking and subsetting.  Visuals accelerate inference or decision; 
tabular detail assures reliable inference or decision.

A communications sage said “Let part stand for the whole.” 
What  can  fit  on  one  sheet  of  paper  or  web  page  is  usually 
sufficient.  My own experience has been that, except when you 
need to provide a comprehensive lookup table, presenting the top 
N observations from a data set is enough, e.g. Rank the Top 10 
Countries in the European Union By Population.

Though  presenting  partial  information,  the  above  graph's 
subtitle  reassures  the  viewer  that  nevertheless  a  significant 
portion of the data has been accounted for.  All of the numbers in 
the titles and footnotes of these graphs are automatically filled in.

Rather than limit the listing to the Top N observations, one 
can instead limit the ranked listing to only enough observations to 
present P% of the total of the measure of interest, e.g. Rank the 
Top Countries That Account for At Least 80% of the European 
Union’s Population.

Here is common code used to prepare input and create some 
of the symbolic variables:

PROC MEANS DATA = work.Input NOPRINT SUM N;
  VAR Measure;
  OUTPUT OUT = work.Summary SUM = Total
         N = Count;
RUN;
DATA _NULL_;
  SET work.Summary;
  CALL SYMPUT('Total', Total);
  CALL SYMPUT('TotalForDisplay',
          TRIM(LEFT(PUT(Total, COMMA13.))));
  CALL SYMPUT('CountForDisplay',
          TRIM(LEFT(PUT(Count, 3.))));
RUN;
PROC SORT DATA = work.Input;
  BY DESCENDING Measure;
RUN; 

Here is common code used for all of the graphs:

%LET FontSize = 32PT;
%LET Font = Verdana;
GOPTIONS RESET = ALL;
GOPTIONS DEVICE = GIF;
GOPTIONS BORDER; /* Put the graph in a box. */
GOPTIONS CBACK = CXFFFF00;
        /* Use RGB Yellow for background. */
GOPTIONS HTEXT = &FontSize. FTEXT = "&Font.";
 /* Height & font for those parts of the graph
    which you do not explicitly assign,
    or for which no controls are available */
GOPTIONS GSFMODE = REPLACE GSFNAME = anyname;
FOOTNOTE1 ANGLE = +90 FONT = NONE
          HEIGHT = 1 ' ';
 /* blank "right-side-note"
    for space at the side of the image. */
FOOTNOTE2 ANGLE = -90 FONT = NONE
          HEIGHT = 1 ' ';
 /* blank "left-side-note"
    for space at the side of the image. */
PATTERN1 V = SOLID COLOR = CX006600;
 /* Use an RGB Medium Dark Green for bars. */
 /* Remove all axis paraphernalia: */
AXIS1 LABEL = NONE MAJOR = NONE MINOR = NONE
      STYLE = 0;

AXIS2 LABEL = NONE MAJOR = NONE MINOR = NONE
      STYLE = 0 VALUE = NONE
      OFFSET = (1PCT, 1PCT);
   /* Offset horizontal axis to insert space
      between bar midpoint label and bar start,
      and between bar end and bar value. */

Here is code for the TopN graph: 

%let MaxN   = 10;
DATA work.TopN;
  RETAIN TotToRpt 0;
  SET work.Input;
  OUTPUT;
  TotToRpt = TotToRpt + Measure;
  IF _N_ = &MaxN.;
  PctOfTotal = (TotToRpt / &Total.) * 100;
  CALL SYMPUT('N_TotToRptForDisplay',
     TRIM(LEFT(PUT(TotToRpt, COMMA13.))));
  CALL SYMPUT('MaxNpctOfTotal',
     TRIM(LEFT(PUT(PctOfTotal, 5.1))));
  STOP;
RUN;
TITLE1 FONT = "&Font."
  COLOR = CX000000   /* RGB black */
  HEIGHT = 3PCT ' ' /* Insert space at top. */
  HEIGHT = &FontSize.
  JUSTIFY = CENTER   /* Force a new line. */
  "Top &MaxN. EU Countries Ranked
   by Population"
  JUSTIFY = CENTER   /* Force a new line. */
  "Account for &MaxNpctOfTotal.% of the Total";
FOOTNOTE3 FONT = "&Font."
  COLOR = CX000000   /* RGB black */
  HEIGHT = &FontSize. JUSTIFY = RIGHT 
  "Subtotal for these 
   countries = &N_TotToRptForDisplay.  ";
FILENAME anyname "C:\TopN.GIF";
PROC GCHART DATA = work.TopN;
  HBAR BarVar / SUMVAR = Measure DISCRETE SUM
                SUMLABEL = NONE DESCENDING
                MAXIS = axis1 RAXIS = axis2;
  FORMAT Measure COMMA13.;
RUN;
QUIT;

Here is code for the TopPct graph: 
%LET MaxPct = 80;
DATA work.TopPct;
  RETAIN TotToRpt 0 TotForPct 0;
  SET work.Input;
  IF _N_ EQ 1 THEN DO;
    TotForPct = (&Total. * &MaxPct.) / 100;
  END;
  OUTPUT;
  TotToRpt = TotToRpt + Measure;
  IF TotToRpt GE TotForPct THEN DO;
    CALL SYMPUT('Pct_TotToRptForDisplay',
         TRIM(LEFT(PUT(TotToRpt, COMMA13.))));
    CALL SYMPUT('Pct_CountToRptForDisplay',
         TRIM(LEFT(PUT(_N_, 3.))));
    STOP;
  END;
RUN;

TITLE1 FONT = "&Font."
  COLOR = CX000000   /* RGB black */
  HEIGHT = 3PCT ' ' /* Insert space at top. */
  HEIGHT = &FontSize.
  JUSTIFY = CENTER   /* Force a new line. */
  "Top &Pct_CountToRptForDisplay. of
   &CountForDisplay. EU Countries
   by Population"
  JUSTIFY = CENTER   /* Force a new line. */
  "Are Enough To Account for"  
  JUSTIFY = CENTER   /* Force a new line. */
  "At Least &MaxPct.%
   of the Total (&TotalForDisplay.)";
FOOTNOTE3 FONT = "&Font."
  COLOR = CX000000   /* RGB black */
  HEIGHT = &FontSize.
  JUSTIFY = RIGHT 
  "Subtotal for these 
   countries = &Pct_TotToRptForDisplay.  ";
FILENAME anyname "C:\TopPct.GIF";
PROC GCHART DATA = work.TopPct;
  HBAR BarVar / SUMVAR=Measure DISCRETE SUM
                SUMLABEL=NONE DESCENDING
                MAXIS=axis1 RAXIS=axis2;
  FORMAT Measure COMMA13.;
RUN;
QUIT;

     LeRoy Bessler, Assurant Health

PRX 5: Matches and replays
[SAS 9] The PRXNEXT function can be called repeatedly to find 
successive  matches  for  a  regular  expression  within  a  source 
string.  In this example we look for all occurrences of  “cat” or 
“mog”. 

DATA test1;
  LENGTH instr $60;
  INPUT instr $ 1-60;
  rx = PRXPARSE("/cat|mog/");
  start = 1;
  stop = LENGTH(TRIM(instr));
  DO UNTIL (p = 0);
    CALL PRXNEXT(rx, start, stop, TRIM(instr),
                 p, 1);
    IF p > 0 THEN DO;
      sub = SUBSTR(instr, p, 1);
      PUT "Matching """ sub +(-1)
          """ found at position " p;
    END;
    ELSE PUT "No more matches found";
  END;
  DATALINES;
Transmogrification of cattle demographics
;

We  initialise  our  START  and  STOP  variables  with  the 
positions  of  the  beginning  and  end  of  the  string,  and  call 
PRXNEXT repeatedly until  no more matches  are  found.   The 
results are:

Matching "mog" found at position 6
Matching "cat" found at position 13
Matching "cat" found at position 23
Matching "mog" found at position 32
No more matches found

Now here is a program that does something slightly different. 
We look for the first occurrence of either  “cat” or  “mog”, and 
then for the first recurrence of that, whichever it was.

DATA test2;
  LENGTH instr $60 p1 p2 $20;
  INPUT instr $ 1-60;
  rx = PRXPARSE("/(cat|mog).*(\1)/");
  CALL PRXSUBSTR(rx, instr, p, 1);
  IF p > 0 THEN DO;
    sub = SUBSTR(instr, p, 1);
    p1 = PRXPOSN(rx, 1, instr);
    p2 = PRXPOSN(rx, 2, instr);
    PUT "Matching substring is: " sub;
    PUT p1 = p2 =;
  END;
  DATALINES;
Transmogrification of cattle demographics
Location of smog-scattering equipment
;

The  new  syntax  here  is  “\1” which  means  “the  text  that 
matched sub-expression 1”. Here we display the entire matching 
sub-string, and also the matches for the two sub-expressions. The 
output from our program is:

Matching substring is: mogrification of cattle 
demog
p1=mog p2=mog
Matching substring is: cation of smog-scat
p1=cat p2=cat

This syntax greatly extends the range of patterns we can look 
for e.g. the regular expression “/([A-Za-z])\1/” matches the first 
doubled letter in a string, though note that in “Llanelli” it would 
match the 2nd pair of “l”s , since the first pair differ in case.

Amadeus Software Consultancy Team

PRX 6: Altering strings
[SAS 9] The  PRXCHANGE call  routine  enables  us  to  modify 
text strings, at times in quite clever ways. It uses a different form 
of regular expression from those we have seen before.  Whereas 
PRXMATCH and the others use expressions of the form:

"/pattern/"
PRXCHANGE regular expressions are of the form:

"s/old/new/"
Actually, in both cases, a character other than “/” can be used 

as delimiter if you prefer.

Here is a simple example, in which we change cats and mogs 
to felines:

DATA test1;
  LENGTH instr outstr $60;
  INPUT instr $ 1-60;
  rx = PRXPARSE("s/cat|mog/feline/");
  CALL PRXCHANGE(rx, 1, instr, outstr);
  PUT "Changed """ instr +(-1) """ to """
                   outstr +(-1) """";
  DATALINES;
magnificat
seismograph
;

The second parameter,  which here we have set  to 1,  is the 
number of times the change is to be applied.  A value of -1 means 
“as many times as possible”.  The output from our program is:

Changed "magnificat" to "magnififeline"
Changed "seismograph" to "seisfelineraph"

What  makes  matters  more  interesting  is  the  ability  to 
substitute  into  a  target  string  the  values  that  matched 
sub-expressions.  Here, for example, is a program that switches 
between British and American date formats:

DATA test2;
  LENGTH instr outstr $60;
  INPUT instr $ 1-60;
  rx = PRXPARSE("s!(\d{2})[/-](\d{2})" ||
                "[/-](\d{2}|\d{4})!$2/$1/$3!");
  CALL PRXCHANGE(rx, 1, instr, outstr);
  PUT "Changed """ instr +(-1) """ to """
                   outstr +(-1) """";
  DATALINES;
25/12/08
04-07-2009
;

The output is:

Changed "25/12/08" to "12/25/08"
Changed "04-07-2009" to "07/04/2009"

Notice first that we have chosen to use “!” as delimiter, since 
we want to use  “/” elsewhere in our expression.  Our first two 
sub-expressions, the parts in parentheses, are two-digit numbers; 
the third is the year, which can be either two or four digits.  These 
sub-expressions are separated by either slashes or hyphens.  In 
the part of our regular expression that builds the new string, we 
can  substitute  in  the  matching  sub-strings  as  $1,  $2  and  $3 
respectively.-

Recall that in a PRXMATCH regular expression, the syntax 
for  referring  to  “the  text  that  matched  sub-expression  1”  was 
“\1”.   When building a new string in a PRXCHANGE regular 
expression,  the syntax  for  the  same thing is  “$1”.   So,  if  we 
wanted to ensure that the separator within the date string, either 
“/” or  “-”, was the same in both places, and reproduce it in the 
output, we could use the slightly fiercer-looking expression:

  "s!(\d{2})([/-])(\d{2})\2(\d{2}|\d{4})" ||
  "!$3$2$1$2$4!"

Here, both “\2” and “$2” appear, in the appropriate parts of 
the expression.

Amadeus Software Consultancy Team

Ask The Consultant
This  part  of  VIEWS  News is  where  you  can  get  your 
technical questions answered.  Send your questions to the 
Editor.

QuickTip - Bubble Plot that has drill down
Q: Is it possible to produce bubble plots with consistent sizes and 
with drill-down?

A: [SAS 9] This  macro will  make bubble plots  with consistent 
sizes across BY groups and to allow drill down (it was based on a 

SAS sample).   The  graph  below was produced  by the macro, 
followed by the macro code and another sample of its use:

%MACRO bubblePlot(data, x, y, size,
                  href=about:blank,
                  font=SWISSB,
                  color=lightblue,
                  labels=0, by=, textSize=4,
                  minSize=, maxSize=);
 %* data - dataset to use for plot;
 %* x - x-axis variable;
 %* y - y-axis variable;
 %* size - variable to use for size of bubbles;
 %* href - drill down link;
 %* font - font for labels on bubbles ;
 %* color - colour of bubbles ;
 %* labels - 1=on each bubble, 0=none;
 %* by - by variable for plot;
 %* textSize - size of labels;
 %* minSize - if specified then sets minimum
              size of bubbles;
 %* maxSize - if specified then sets maximum
              size of bubbles;
 %IF &minSize. > AND &maxSize. > %THEN %DO;
 PROC SQL NOPRINT;
   SELECT MIN(&size.), MAX(&size.),
          RANGE(&size.)   
   INTO   :realMin, :realMax, :realRange
   FROM   &data.;
 QUIT;
 %PUT _LOCAL_;
 %END;
 * Create Annotate data set, Anno:
   HSYS='1' specifies annotate to use a
   SIZE unit of percent of plot area;
 DATA _anno;
   LENGTH function text $8 style $32 html $128;
   RETAIN xsys ysys '2' hsys '1' when 'A'
          color "&color.";
   SET &data. (RENAME = (&x. = x &y. = y));
 %IF &by. > %THEN
   BY &by.;
 ;
   function = 'PIE';
   style = 'PSOLID';

 %IF &minSize. > & &maxSize. > %THEN %DO;
   pct = (&size. - &realMin.) / &realRange.;
   range = &maxSize. - &minSize.;
   size = &minSize. + pct * range;
 %END;
 %ELSE %DO;
   size = &size.;
 %END;
   angle = 0;
   rotate = 360;
   text = '';
   html = "title=""" || STRIP(VLABEL(x)) || 
      " = " || STRIP(PUTN(x, VFORMAT(x))) ||
      '0D'x || STRIP(VLABEL(y)) || " = " ||
      STRIP(PUTN(y, VFORMAT(y))) || '0D'x ||
      STRIP(VLABEL(size)) || " = " ||
      STRIP(PUTN(&size., VFORMAT(&size.))) ||
      """ href=""&href.""";
   OUTPUT;
 %IF &labels. %THEN %DO;
   function = 'LABEL'; 
   position = '1'; 
   line = 0;
   style = "&font.";
   text = STRIP(&size.);
   size = &textSize.;
   OUTPUT;
 %END;
 RUN;
 SYMBOL I = NONE V = NONE C = WHITE R = 10;
 AXIS1 MINOR = NONE OFFSET = (2, 2);
 AXIS2 MINOR = NONE OFFSET = (3, 3);
 * Produce the plot;
 PROC GPLOT DATA = _anno;
 %IF &by. > %THEN
   BY &by.;
 ;
 * use GPLOT to get axes and graph area setup
   so we can annotate on the bubbles and labels
   - including links on bubbles;
   PLOT y * x / VAXIS = axis1 HAXIS = axis2
                ANNO = _anno;
 RUN;
 QUIT;
%MEND bubblePlot;
* Setup the graphics environment;
GOPTIONS RESET = ALL DEVICE = SASEMF
                   /* sasemf/activex/java */
         FTEXT = SWISS FTITLE = SWISS
         HTEXT = 3PCT HTITLE = 10PCT BORDER;
ODS _ALL_ CLOSE;
ODS HTML;
TITLE 'height vs. weight = age';
%bubblePlot(sashelp.class, height, weight, age,
            minsize = 1, maxsize = 3);

Here  is  another  graph  and  sample program which uses  the 
macro:

PROC FORMAT;
  VALUE ae
    1='Cough' 2='Death' 3='Flu' other='Other';
RUN;
DATA aeonset (KEEP = ae days count);
  LABEL ae = 'Adverse Events' days = 'Days'
        count = 'Count';
  FORMAT ae ae.;
  DO I = 1 TO 25;
    ae = ROUND(RANUNI(1) * 3);
    days = ROUND(RANUNI(1) * 50);
    count = ROUND(RANUNI(1) * 100);
    OUTPUT;
  END;
RUN;
TITLE 'Random Adverse Events onset';
%bubblePlot(aeonset, days, ae, count,
            minsize = 1, maxsize = 3);
ODS _ALL_ CLOSE;

Phil Mason, Wood Street Consultants Ltd
Taken from Phil Mason's SAS Tips.

Formats, Options, and Functions
This  section  is  devoted  to  the  description  of  useful,  or 
unusual, SAS Formats, Options, and Functions.

[SAS 9.2] The  BESTDw.p format  displays  numeric  values, 
lining up  decimal  places  for  values  of  similar  magnitude,  and 
displays integers without decimal places.  The “p” in the syntax 
optionally specifies the precision, defaulting to 3 if omitted. 

[SAS 9.2] It is now possible to set a default record length for 
external  files  written  or  read  by SAS using the  system option 
LRECL=.  The default value is 256.

[SAS 9.2] The  CALL  SORTC() and  CALL  SORTN() 
routines  can  be  used  to  sort  character  and  numeric  variables, 
respectively.

PhUSE News
2008 PhUSE Conference
The 2008 PhUSE conference was held in Manchester from 12-15 
October.  With more than 300 participants from 16 countries and 
100 companies,  PhUSE is now the biggest  event for statistical 
programmers in Europe.  There were more than 100 papers and 
posters  presented  including the  usual  technical  ones  but  more 
than 15 papers around this year's  theme "Working in a Global 
Environment".   All  papers  are  now  available  on  the  PhUSE 
website,  www.phuse.eu,  and  on  Lex  Jansen's  website, 
www.lexjansen.com.

Next year's conference will be held in Basel, Switzerland on 
19-21 October and will be chaired by Benjamin Szilagyi (Roche). 
We are  currently planning the  event,  including pre-conference 
training courses on Sunday 18 October.  More information will 
be available early in 2009.

Thanks for having contributed to the success of PhUSE.  See 
you next year in Basel.

Olivier Leconte, 2008 PhUSE Conference Chair

News
SAS Forum UK 2008:   Business  Analytics  - 
The Bridge to Innovation
On  21-22  October  2008  over  400  SAS users  from over  120 
organisations gathered in Telford for this year's SAS Forum.  The 
two day event featured key note presentations from futurologist 
Dr Ian  Pearson,  Anne  Milley  from  SAS,  Zoe  and  Sky  from 
Wildtrack, CTO Steve Brobst from Teradata and Jon Tickle of 
Big Brother and Brainiac fame.  Delegates also had the chance to 
choose from technology and industry themed break-out sessions 
as well as look around the exhibition.

Of course, SAS Forum is also renowned for its evening party, 
and  this  year's  theme was “The Wild  West”.   Feedback  from 
attendees has been good, but we're always looking for ways to 
improve, so if you have any ideas or feedback please email Phil 
Male at philip.male@suk.sas.com.

Phil Male, SAS UK

In Brief
• Back issues  of  VIEWS News are  available  from the 

VIEWS web site, and also from the VIEWS News page 
at:
  www.sasCommunity.org/wiki/VIEWS_News

• Phil Mason now has a SAS tips newsletter, which can 
be  requested  by  going to  www.woodstreet.org.uk 
and then clicking on Tips.  Previous tips can also be 
found on the web site.

• More SAS hints and tips  can be found on the SAS 
FAQ pages  www.hollandnumerics.com/sasfaq and 
also  on  the  Amadeus  Software  Tips  pages  at 
www.amadeus.co.uk/TipsList.aspx?ST=1.

• Looking for  a  SAS-related  job?   Try the  following 
web pages:  www.hollandnumerics.com/careers and 
www.amadeus.co.uk/Careers.aspx.

• You are invited to join the SAS Professionals social 
networking group, described as the “365 Day SIG” for 
SAS Users in the UK, at:
   sasprofessionals.ning.com

• Phil Holland has also started a group page on Squidoo 
for  other  pages  related  to  SAS,  and  would  like  to 
make it  a  central  resource  for SAS hints, tips,  links 
and books, at:
 www.squidoo.com/groups/SAS_users

Stop Press!
New SAS-related book by Philip R Holland
Phil Holland has just published his 2nd SAS-related book, called 
“Power User's Guide to SAS Programming”.  This book brings 
together a wealth of ideas about strategic and tactical solutions to 
everyday  situations  experienced  when  transferring,  extracting, 
processing  and  reporting  data.  Most  of  the  solutions  can  be 
achieved using the SAS components you already license, so you 
won't need to throw money at the problems.

The book is available now as a print-on-demand book, or as a 
download, from stores.lulu.com/hollandnumerics.  However, it 
is hoped it  will also be available through Amazon in the New 
Year.

Philip R Holland

Diary
Are you organising an event that would be of interest to the  
VIEWS readership?  Let us know as we are interested in all  
non-profit making events related to SAS.

December 2008
11 SAS9 Business Intelligence-Delivered, Oxfordshire

See the Amadeus Software web site  www.amadeus.co.uk 
for further details.

February 2009
10 SAS Programming Tips and Tricks, Oxfordshire

See the Amadeus Software web site  www.amadeus.co.uk 
for further details.
25 New and Enhanced Features of SAS 9.2, Oxfordshire

See the Amadeus Software web site  www.amadeus.co.uk 
for further details.

March 2009
12 From Data to Knowledge: A Journey into SAS9, 

Oxfordshire
See the Amadeus Software web site  www.amadeus.co.uk 

for further details.
22-25 SAS Global Forum, National Harbor, MD, USA

The SAS Global  Forum (SGF,  formerly called SUGI)  is 
held each year  and focuses on the technical  aspects of SAS 
software.  See www.sasglobalforum.org for more details.
26-27 Best Practices in SAS Statistical Programming in 

Regulatory Submission, New Brunswick, NJ, USA
These courses are intended for anyone directly or indirectly 

responsible for the creation, content or validation of summary 
tables, data lists and graphs used to support research, drug or 
medical device efficacy and safety in a regulatory submission. 
See tinyurl.com/5ayrmq for more details of the courses from 
Sunil Gupta.

June 2009
1 SAS Programming Tips and Tricks, Oxfordshire

See the Amadeus Software web site  www.amadeus.co.uk 
for further details.
25 New and Enhanced Features of SAS 9.2, Oxfordshire

See the Amadeus Software web site  www.amadeus.co.uk 
for further details.

July 2009
22 From Data to Knowledge: A Journey into SAS9, 

Oxfordshire
See the Amadeus Software web site  www.amadeus.co.uk 

for further details.

September 2009
24 SAS Programming Tips and Tricks, Oxfordshire

See the Amadeus Software web site  www.amadeus.co.uk 
for further details.

October 2009
19-21 PhUSE Conference, Basel, Switzerland

The  Pharmaceutical  Users  Software  Exchange  (PhUSE) 
conference  is  the  premier  programming  event  for  the 
pharmaceutical  industry in Europe.   See  www.phuse.eu for 
more details.
28 New and Enhanced Features of SAS 9.2, Oxfordshire

See the Amadeus Software web site  www.amadeus.co.uk 
for further details.

November 2009
19 From Data to Knowledge: A Journey into SAS9, 

Oxfordshire
See the Amadeus Software web site  www.amadeus.co.uk 

for further details.

December 2009
1 SAS Programming Tips and Tricks, Oxfordshire

See the Amadeus Software web site  www.amadeus.co.uk 
for further details.

All VIEWS event information will be posted and/or emailed 
to registered members of VIEWS, and will include an event 
application  form.   Please  send  your  queries  about  any 

